
ANSWERS

..Welcome to the practical sessions of Reflection and Context-Oriented Programming. These
sessions are based on Pharo, an open source implementation of the Smalltalk programming
language. Pharo can be downloaded from pharo-project.org; at the moment we recommend
using version 2.0. Even though it is still tagged as ‘unstable’, it should work fine for the
purposes of these sessions.

There is plenty of documentation in the book Pharo by Example, which you can freely
download from pharobyexample.org. If at any point in time you have doubts on how to use
the system, it is likely that the answer lies somewhere in that book. Please use it as a reference
for the practical sessions (and it might help as a complement to some of the theory sessions
too).

Conventions Questions marked with : require more insight about the language and its im-
plementation than the others, and are therefore optional.

Basic Expressions

1. Evaluate each of the following expressions in a workspace and inspect their result. What is
the class of each result?

Expression Result Class of result

2001 2001 SmallInteger

2 raisedTo: 30 1073741824 LargePositiveInteger

15 / 25 (3/5) Fraction

3.1415 3.1415 Float

$a $a Character

1 @ 2 1@2 Point

’Hello Smalltalk’ ’Hello Smalltalk’ ByteString

#foobar #foobar ByteSymbol

#’a/b’ #’a/b’ ByteSymbol

#’2’ #’2’ ByteSymbol

Expression Result Class of result

#2 2 SmallInteger

#(1 2 3) #(1 2 3) Array

#(1 + 2. 3 squared.) #(1 #+ 2 #’.’ 3 #squared #’.’)) Array

{1 + 2. 3 squared.} #(3 9) Array

true true True

nil nil Unde nedObject

Float Float Float class

Note that the #’symbol syntax using strings’ allows the creation of symbols with arbitrary names,
containing spaces and any other special character as part of the symbol’s name. For a complete overview
of the syntax, refer to Pharo by Example.

2. : In exercise 1, why is the class of the second expression different from that of the first one?
Could this result vary among Smalltalk implementations?
The reason why 2 raisedTo: 30 is a LargePositiveInteger rather than a SmallInteger, even though
the result normally ts in a 32-bit word, is that 2 bits are reserved by the virtual machine to handle tagged
pointers.

Message Expressions

For each of the following expressions, answer the questions.

3. 3 + 4

What is the receiver object? 3
What is the message selector? #+
What are the arguments? 4
What is the message? + 4
What is the result? 7

4. Date today

What is the receiver object? Date
What is the message selector? #today
What are the arguments? no arguments
What is the message? today
What is the result? 14 February 2011

5. anArray := #(1 2).
anArray at: 1 put: ’hello’

Consider the second expression.
What is the receiver object? the array object stored in the “anArray” variable

What is the message selector? #at:put:
What are the arguments? 1 and ’hello’
What is the message? at: 1 put: ’hello’
What is the result? ’hello’
What is the value of anArray after evaluation? #(’hello’ 2)

6. ’Hello’, ’Smalltalk’

What is the receiver object? ’Hello’
What is the message selector? #,
What are the arguments? ’Smalltalk’
What is the message? , ’Smalltalk’
What is the result? ’HelloSmalltalk’

Message Precedence

7. Evaluate the following expressions and inspect their results. Why are they different?
3 + 4 * 2 → 14
3 + (4 * 2) → 11

In Smalltalk, binary operators such as + and * are normal messages. As a consequence, these messages
follow the normal Smalltalk message precedence rules, irrespective of their implementation (i.e. whether
they are summation, multiplication, string concatenation, and so forth).

8. Which sets of parentheses are redundant in the following expressions?
((3 + 4) + (2 * 2) + (2 * 3))
(x isZero) ifTrue: [...]
(x includes: y) ifTrue: [...]

3 + 4 + (2 * 2) + (2 * 3)
x isZero ifTrue: [...]
(x includes: y) ifTrue: [...]

9. Examine the following expressions and write down in which order messages are sent to what
objects in the following expressions:

a) Date today daysInMonth

A: Date today
The message today is sent to the class Date.

B: A daysInMonth
The message daysInMonth is sent to the result of expression A.

b) Transcript show: (45 + 9) printString

A: 45 + 9
The message + 9 is sent to 45.

B: A printString
The message printString is sent to the result of expression A.

C: Transcript show: B
The result of expression B is the argument passed with show:

c) 5 @ 5 extent: 6.0 truncated @ 7

A: 6.0 truncated
The message truncated is sent to 6.0.

B: 5 @ 5
The message @ 5 is sent to 5.

C: A @ 7
The message @ 7 is sent to the result of expression A

D: B extent: C
The message with selector #extent: and argument resulting from expression C is sent to the result
of expression B.

Symbols, Strings and Comparisons

10. Evaluate the following expressions and write down their result. What is the meaning of the =
and == messages?

a) ’ab’ = (’a’, ’b’) true

b) ’ab’ == (’a’, ’b’) false

c) ’ab’ == ’ab’ true

d) #thisIsASymbol == #thisIsASymbol true

e) #thisIsASymbol == #thisisasymbol false

f) #’hey you!’ == ’hey you!’ asSymbol true

The = message performs an object equivalence comparison: it decides whether the receiver is equivalent
to the given argument, for some domain-speci c de nition of “equivalence” (e.g. two strings are equiv-
alent if they have the same characters at the same positions). The ==message performs an object identity
comparison: it compares if the receiver is identical to the argument; this relationship does not depend
on the speci c nature (domain) of the involved objects.

Note that Smalltalk’s = operator is not an assignment as is the case in Java. In Smalltalk, assignment is
performed through the := operator.

11. Why are the parentheses in expressions a) and b) needed?
To override left-to-right precedence of binary operators such as equality comparison (=) and string con-
catenation (,).

12. Why do the results of expressions a) and b) differ?
Expression a) compares the content of both strings, which is the same. Expression b) compares the identity
of both strings, which are different objects (even though they have the same content).

13. : Why do the results of expressions b) and c) differ? Could this result vary among Smalltalk
implementations?
In both expressions b) and c), the string to the left of == is a literal string; the language parser reads the lit-
eral representation and creates a corresponding ByteString instance. Let’s consider now the string to the
right of the == operator. For expression c), it is the same literal string. The parser knows it has already read
this literal while compiling the current expression, and as an optimisation, decides to reuse the already
created ByteString object. For Smalltalk implementations which do not perform this optimisation, the
result of expression c) will be false.

14. Can you explain the result of expression d)?
Symbols with the same name are one and the same object. The Smalltalk parser takes care of returning
the same symbol instance each time it nds a symbol literal with the same name.

15. Can you explain the result of expression f)?
The asSymbol message looks up the dictionary of registered symbols to see if a symbol with the given
name exists, creates it if non existent, and returns it. In this particular example, the symbol will already
exist by the time asSymbol is sent, because the parser creates it when it reads the symbol literal #’hey
you’.

16. What is a symbol? Define as precisely as possible.

Code Blocks

Evaluate the following pieces of Smalltalk code involving blocks. Write down the result of each
piece. Can you explain the results?

17. [1 + 1] → [1 + 1], of type BlockClosure
A block is an object that represents a closure —that is, code that can be executed, and the lexical (i.e.
textual) scope that surrounds the block expression.

18. [1 + 1] value → 2
The message value evaluates the unparametrised block given as receiver.

19. |aBlock|
aBlock := [1 + 1].
aBlock value. → 2
aBlock value inspect. → 2

Blocks can be assigned to variables and can be executed multiple times.

20. |blockA blockB|
blockA := [:arg | arg + 1].
blockB := [:arg1 :arg2 | arg1 + arg2].
(blockA value: 2) + (blockB value: 3 value: 2) → 8

Blocks can have arguments and can be evaluated with the message
value:<arg1> value:<arg2> ... value: <argn>.

Conditionals and Iteration

Blocks are essential in Smalltalk’s implementation of common control-flow expressions. Be-
cause Smalltalk does not have special syntax for these, they are implemented as expressions
that operate on blocks. Answer the following questions and verify your answer by executing
the corresponding Smalltalk expressions.

21. Will the evaluation of the following expressions yield the same result? Why (not)?
3 < 2 ifTrue: [’yes’] ifFalse: [’no’]
3 < 2 ifTrue: [Transcript show: ’yes’] ifFalse: [’no’]
3 < 2 ifTrue: (Transcript show: ’yes’) ifFalse: ’no’
3 > 2 ifFalse: [’no’]

The rst two expressions return ’no’ and there are no side-effects on Transcript because the “true block”
is never evaluated. In the third expression, the rst argument expression prints ’yes’ on the Transcript
as a side effect, though the conditional still yields ’no’ as in the previous case. The last expression returns
nil; the “false block” passed as argument is never evaluated.

22. Write a corresponding piece of pseudo-Java code (using Java control-flow constructs) for the
following Smalltalk expressions:
Note that Java Lambda Expressions, recently introduced in JDK7, could also be used in the answers given
below. This is left as a further exercise to the reader :-)

a) i := 1.
[i < 10] whileTrue: [Transcript show: i printString. i := i + 1]

int i = 1;
while (i < 10)

{ System.out.print(i); i++; }

b) (1 to: 9) do: [:x | Transcript show: x printString]

for(int i = 1; i < 10 ; i++)
System.out.print(i);

c) (1 to: 9)
do: [:x | Transcript show: x printString]
separatedBy: [Transcript cr]

for(int i = 1; i < 10 ; i++) {
System.out.print(i);
if(i != 9) System.out.println();

}

d) #(21 23 53 66 87) collect:[:x | x * 2] → #(42 46 106 132 174)

int[] array = { 21, 23, 53, 66, 87 };
int[] result = new int[array.length];
for(int i = 0; i < array.length ; i++)

result[i] = array[i] * 2;

e) #(21 23 53 66 87) detect: [:x | x odd] → 21

int[] array = {21, 23, 53, 66, 87};
Integer result = null;
for (int i = 0; i < array.length ; i++) {

if (array[i] % 2 == 1) {
result = array[i];
break;

}
}
if (result == null)

throw new Exception(”Not found”);

23. What is the final result of sum in the following Smalltalk program? → 250

| array sum |
sum := 0.
array := #(21 23 53 66 87).
array do: [:item | sum := sum + item].
sum

24. How can the above piece of code be rewritten to use explicit array indexing (with the at:
message) to access the array elements? Test your version.
| array sum |
sum := 0.
array := #(21 23 53 66 87).
1 to: array size do: [:i | sum := sum + (array at: i)].
sum

Note that arrays are 1-based –that is, the rst valid index is 1, rather than 0.

25. Rewrite your code using the inject:into: message.
| array sum |
array := #(21 23 53 66 87).
sum := array inject: 0 into: [:subtotal :item | subtotal + item].
sum

Classes and Objects

26. Use the Smalltalk browser to create a class Counter. Add the necessary instance variables and
methods to the class such that the following piece of Smalltalk code can be executed:

1 |counterA counterB|
2 counterA := Counter new.
3 counterA increment; increment.
4 counterA value inspect.
5 counterA incrementBy: 4.
6 counterA do: [:c | Transcript show: c printString].
7 counterA decrement.
8 counterA decrementBy: 2.
9 counterB := Counter startingAt: 10.

10 (counterA + counterB) inspect

� Note that you need not add a method new to the class-side of Counter, because you can use the one that is
already defined for all classes in the system. This default implementation will create a new instance of the class,
send it the initialize message, and return this freshly initialised instance as result. Hence, to perform any
initialisation chores, you just need to override the initialize instance method of Counter. This is the standard
protocol Smalltalk follows for object creation and initialisation.

The solution code is available as a Monticello package, called R+COP-Solutions-Lab1. Monticello is
Pharo’s main package management system. For usage information, please consult Pharo by Example.

27. What does the semicolon (;) on line 3 of exercise 26 mean?
It indicates a cascaded message. The second increment message on line 3 is sent to the same receiver the
previous message had, which in this case is counterA.

As another example, in the expression 4 + 5; + 2 the message + 5 is sent to 4, and then the message +
2 is sent to 4.

Cascaded messages are always sent to the receiver of the rst message in the cascade. For illustration,
consider the following expressions:

• 5 squared; squared; squared → 25

• 5 squared squared; squared → 625

• 5 squared squared; + 1 → 26

28. Load the package R+COP-Lab1 into your Pharo image. The package contains a battery of unit
tests that you can apply to check that class Counter from exercise 26 works correctly —though
the actual goal of this exercise is rather to get you familiarised with Monticello, the package
management system of Pharo, and also with SUnit, the unit testing framework of Pharo. You
can read all needed information in Chapters 6 and 7 of Pharo by Example.

